
NAG Library Routine Document

D02KAF

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

D02KAF finds a specified eigenvalue of a regular second-order Sturm–Liouville system defined on a finite
range, using a Pruefer transformation and a shooting method.

2 Specification

SUBROUTINE D02KAF (XL, XR, COEFFN, BCOND, K, TOL, ELAM, DELAM, MONIT,
IFAIL)

&

INTEGER K, IFAIL

REAL (KIND=nag_wp) XL, XR, BCOND(3,2), TOL, ELAM, DELAM

EXTERNAL COEFFN, MONIT

3 Description

D02KAF finds a specified eigenvalue ~� of a Sturm–Liouville system defined by a self-adjoint differential
equation of the second-order

p xð Þy0
� �0 þ q x;�ð Þy ¼ 0, a < x < b,

together with boundary conditions of the form

a2y að Þ ¼ a1p að Þy0 að Þ

b2y bð Þ ¼ b1p bð Þy0 bð Þ
at the two, finite, end points a and b. The functions p and q, which are real-valued, are defined by
COEFFN.

For the theoretical basis of the numerical method to be valid, the following conditions should hold on the
coefficient functions:

(a) p xð Þ must be nonzero and must not change sign throughout the closed interval a; b½ �;

(b)
@q
@�

must not change sign and must be nonzero throughout the open interval a; bð Þ and for all relevant

values of �, and must not be identically zero as x varies, for any relevant value �; and,

(c) p and q should (as functions of x) have continuous derivatives, preferably up to the fourth-order, on
a; b½ �. The differential equation code used will integrate through mild discontinuities, but probably

with severely reduced efficiency. Therefore, if p and q violate this condition, D02KDF should be
used.

The eigenvalue ~� is determined by a shooting method based on a Pruefer transformation of the differential

equations. Providing certain assumptions are met, the computed value of ~� will be correct to within a
mixed absolute/relative error specified by TOL. D02KAF is a driver routine for the more complicated
routine D02KDF whose specification provides more details of the techniques used.

A good account of the theory of Sturm–Liouville systems, with some description of Pruefer
transformations, is given in Chapter X of Birkhoff and Rota (1962). An introduction to the use of
Pruefer transformations for the numerical solution of eigenvalue problems arising from physics and
chemistry is given in Bailey (1966).

4 References

Bailey P B (1966) Sturm–Liouville eigenvalues via a phase function SIAM J. Appl. Math. 14 242–249

Birkhoff G and Rota G C (1962) Ordinary Differential Equations Ginn & Co., Boston and New York

5 Parameters

1: XL – REAL (KIND=nag_wp) Input
2: XR – REAL (KIND=nag_wp) Input

On entry: the left- and right-hand end points a and b respectively, of the interval of definition of the
problem.

Constraint: XL < XR.

3: COEFFN – SUBROUTINE, supplied by the user. External Procedure

COEFFN must compute the values of the coefficient functions p xð Þ and q x;�ð Þ for given values of
x and �. Section 3 states the conditions which p and q must satisfy.

The specification of COEFFN is:

SUBROUTINE COEFFN (P, Q, DQDL, X, ELAM, JINT)

INTEGER JINT

REAL (KIND=nag_wp) P, Q, DQDL, X, ELAM

1: P – REAL (KIND=nag_wp) Output

On exit: the value of p xð Þ for the current value of x.

2: Q – REAL (KIND=nag_wp) Output

On exit: the value of q x;�ð Þ for the current value of x and the current trial value of �.

3: DQDL – REAL (KIND=nag_wp) Output

On exit: the value of
@q
@�

x;�ð Þ for the current value of x and the current trial value of �.

However DQDL is only used in error estimation and, in the rare cases where it may be
difficult to evaluate, an approximation (say to within 20%) will suffice.

4: X – REAL (KIND=nag_wp) Input

On entry: the current value of x.

5: ELAM – REAL (KIND=nag_wp) Input

On entry: the current trial value of the eigenvalue parameter �.

6: JINT – INTEGER Input

This parameter is included for compatibility with the more complex routine D02KDF
(which is called by D02KAF).

On entry: need not be set.

COEFFN must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D02KAF is called. Parameters denoted as Input must not be changed by
this procedure.

4: BCONDð3,2Þ – REAL (KIND=nag_wp) array Input/Output

On entry: BCONDð1; 1Þ and BCONDð2; 1Þ must contain the numbers a1, a2 specifying the left-
hand boundary condition in the form

a2y að Þ ¼ a1p að Þy0 að Þ
where a2j j þ a1p að Þj j 6¼ 0.

BCONDð1; 2Þ and BCONDð2; 2Þ must contain b1, b2 such that

b2y bð Þ ¼ b1p bð Þy0 bð Þ
where b2j j þ b1p bð Þj j 6¼ 0.

Note the occurrence of p að Þ, p bð Þ in these formulae.

On exit: BCONDð3; 1Þ and BCONDð3; 2Þ hold values �l; �r estimating the sensitivity of the
computed eigenvalue to changes in the boundary conditions. These values should only be of
interest if the boundary conditions are, in some sense, an approximation to some ‘true’ boundary
conditions. For example, if the range [XL, XR] should really be 0;1½ � but instead XR has been
given a large value and the boundary conditions at infinity applied at XR, then the sensitivity
parameter �r may be of interest. Refer to Section 8.5 in D02KDF, for the actual meaning of �r and
�l.

5: K – INTEGER Input

On entry: k, the index of the required eigenvalue when the eigenvalues are ordered

�0 < �1 < �2 < � � � < �k < � � � .
Constraint: K � 0.

6: TOL – REAL (KIND=nag_wp) Input

On entry: the tolerance parameter which determines the accuracy of the computed eigenvalue. The
error estimate held in DELAM on exit satisfies the mixed absolute/relative error test

DELAM � TOL�max 1:0; ELAMj jð Þ, ð1Þ
where ELAM is the final estimate of the eigenvalue. DELAM is usually somewhat smaller than the
right-hand side of (1) but not several orders of magnitude smaller.

Constraint: TOL > 0:0.

7: ELAM – REAL (KIND=nag_wp) Input/Output

On entry: an initial estimate of the eigenvalue ~�.

On exit: the final computed estimate, whether or not an error occurred.

8: DELAM – REAL (KIND=nag_wp) Input/Output

On entry: an indication of the scale of the problem in the �-direction. DELAM holds the initial
‘search step’ (positive or negative). Its value is not critical, but the first two trial evaluations are
made at ELAM and ELAMþ DELAM, so the routine will work most efficiently if the eigenvalue
lies between these values. A reasonable choice (if a closer bound is not known) is about half the
distance between adjacent eigenvalues in the neighbourhood of the one sought. In practice, there
will often be a problem, similar to the one in hand but with known eigenvalues, which will help one
to choose initial values for ELAM and DELAM.

If DELAM ¼ 0:0 on entry, it is given the default value of 0:25�max 1:0; ELAMj jð Þ.
On exit: if IFAIL ¼ 0, DELAM holds an estimate of the absolute error in the computed eigenvalue,

that is ~�� ELAM
�� �� ’ DELAM, where ~� is the true eigenvalue.

If IFAIL 6¼ 0, DELAM may hold an estimate of the error, or its initial value, depending on the value
of IFAIL. See Section 6 for further details.

9: MONIT – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

MONIT is called by D02KAF at the end of each iteration for � and allows you to monitor the
course of the computation by printing out the parameters (see Section 9 for an example).

If no monitoring is required, the dummy (sub)program D02KAY may be used. (D02KAY is
included in the NAG Library.)

The specification of MONIT is:

SUBROUTINE MONIT (NIT, IFLAG, ELAM, FINFO)

INTEGER NIT, IFLAG

REAL (KIND=nag_wp) ELAM, FINFO(15)

1: NIT – INTEGER Input

On entry: 15 minus the number of iterations used so far in the search for ~�. (Up to 15
iterations are permitted.)

2: IFLAG – INTEGER Input

On entry: describes what phase the computation is in.

IFLAG < 0

An error occurred in the computation at this iteration; an error exit from D02KAF
will follow.

IFLAG ¼ 1

The routine is trying to bracket the eigenvalue ~�.

IFLAG ¼ 2

The routine is converging to the eigenvalue ~� (having already bracketed it).

Normally, the iteration will terminate after a sequence of iterates with IFLAG ¼ 2, but

occasionally the bracket on ~� thus determined will not be sufficiently small and the
iteration will be repeated with tighter accuracy control.

3: ELAM – REAL (KIND=nag_wp) Input

On entry: the current trial value of ~�.

4: FINFOð15Þ – REAL (KIND=nag_wp) array Input

On entry: information about the behaviour of the shooting method, and diagnostic
information in the case of errors. It should not normally be printed in full if no error has
occurred (that is, if IFLAG � 0), though the first few components may be of interest to
you. In case of an error (IFLAG < 0) all the components of FINFO should be printed.

The contents of FINFO are as follows:

FINFOð1Þ
The current value of the ‘miss-distance’ or ‘residual’ function f �ð Þ on which the

shooting method is based. f ~�
� �
¼ 0 in theory. This is set to zero if IFLAG < 0.

FINFOð2Þ
An estimate of the quantity @� defined as follows. Consider the perturbation in the
miss-distance f �ð Þ that would result if the local error in the solution of the
differential equation were always positive and equal to its maximum permitted
value. Then @� is the perturbation in � that would have the same effect on f �ð Þ.
Thus, at the zero of f �ð Þ; @�j j is an approximate bound on the perturbation of the
zero (that is the eigenvalue) caused by errors in numerical solution. If @� is very
large then it is possible that there has been a programming error in COEFFN such
that q is independent of �. If this is the case, an error exit with IFAIL ¼ 5 should

follow. FINFOð2Þ is set to zero if IFLAG < 0.

FINFOð3Þ
The number of internal iterations, using the same value of � and tighter accuracy
tolerances, needed to bring the accuracy (that is, the value of @�) to an acceptable
value. Its value should normally be 1:0, and should almost never exceed 2:0.

FINFOð4Þ
The number of calls to COEFFN at this iteration.

FINFOð5Þ
The number of successful steps taken by the internal differential equation solver at
this iteration. A step is successful if it is used to advance the integration.

FINFOð6Þ
The number of unsuccessful steps used by the internal integrator at this iteration.

FINFOð7Þ
The number of successful steps at the maximum step size taken by the internal
integrator at this iteration.

FINFOð8Þ
Not used.

FINFOð9Þ to FINFOð15Þ
Set to zero, unless IFLAG < 0 in which case they hold the following values
describing the point of failure:

FINFOð9Þ
1 or 2 depending on whether integration was in a forward or backward
direction at the time of failure.

FINFOð10Þ
The value of the independent variable, x, the point at which the error
occurred.

FINFOð11Þ, FINFOð12Þ, FINFOð13Þ
The current values of the Pruefer dependent variables �, � and �
respectively. See Section 3 in D02KEF for a description of these variables.

FINFOð14Þ
The local-error tolerance being used by the internal integrator at the point of
failure.

FINFOð15Þ
The last integration mesh point.

MONIT must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D02KAF is called. Parameters denoted as Input must not be changed by
this procedure.

10: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this parameter you should
refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit (as
defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, K < 0,
or TOL � 0:0.

IFAIL ¼ 2

On entry, a1 ¼ p að Þa2 ¼ 0,
or b1 ¼ p bð Þb2 ¼ 0,

(the array BCOND has been set up incorrectly).

IFAIL ¼ 3

At some point between XL and XR the value of p xð Þ computed by COEFFN became zero or
changed sign. See the last call of MONIT for details.

IFAIL ¼ 4

After 15 iterations the eigenvalue had not been found to the required accuracy.

IFAIL ¼ 5

The ‘bracketing’ phase (with IFLAG of the MONIT equal to 1) failed to bracket the eigenvalue
within ten iterations. This is caused by an error in formulating the problem (for example, q is
independent of �), or by very poor initial estimates of ELAM and DELAM.

On exit, ELAM and ELAMþ DELAM give the end points of the interval within which no
eigenvalue was located by the routine.

IFAIL ¼ 6

To obtain the desired accuracy the local error tolerance was set so small at the start of some sub-
interval that the differential equation solver could not choose an initial step size large enough to
make significant progress. See the last call of MONIT for diagnostics.

IFAIL ¼ 7

At some point the step size in the differential equation solver was reduced to a value too small to
make significant progress (for the same reasons as with IFAIL ¼ 6). This could be due to
pathological behaviour of p xð Þ and q x;�ð Þ or to an unreasonable accuracy requirement or to the
current value of � making the equations ‘stiff’. See the last call of MONIT for details.

IFAIL ¼ 8

TOL is too small for the problem being solved and the machine precision being used. The local

value of ELAM should be a very good approximation to the eigenvalue ~�.

IFAIL ¼ 9

C05AZF, called by D02KAF, has terminated with the error exit corresponding to a pole of the
matching function. This error exit should not occur, but if it does, try solving the problem again
with a smaller value for TOL.

Note: error exits with IFAIL ¼ 2, 3, 6, 7 or 9 are caused by the inability to set up or solve the differential
equation at some iteration and will be immediately preceded by a call of MONIT giving diagnostic
information.

IFAIL ¼ 10
IFAIL ¼ 11
IFAIL ¼ 12

A serious error has occurred in an internal call to an interpolation routine. Check all (sub)program
calls and array dimensions. Seek expert help.

7 Accuracy

The absolute accuracy of the computed eigenvalue is usually within a mixed absolute/relative bound
defined by TOL (as defined above).

8 Further Comments

The time taken by D02KAF depends on the complexity of the coefficient functions, whether they or their
derivatives are rapidly changing, the tolerance demanded, and how many iterations are needed to obtain
convergence. The amount of work per iteration is roughly doubled when TOL is divided by 16. To make
the most economical use of the routine, one should try to obtain good initial values for ELAM and
DELAM.

See Section 8 in D02KDF for a discussion of the technique used.

9 Example

This example finds the fourth eigenvalue of Mathieu’s equations

y00 þ �� 2q cos 2xð Þy ¼ 0

with boundary conditions

y0 0ð Þ ¼ y0 �ð Þ ¼ 0

and q ¼ 5. We use a starting value ELAM ¼ 15:0 and a step DELAM ¼ 4:0. We illustrate the effect of
varying TOL by choosing TOL ¼ 1.0E�5 and 1.0E�6 (note the change in the output value of the error
estimate DELAM). The range of integration and initial estimates are read from a data file.

9.1 Program Text

! D02KAF Example Program Text
! Mark 24 Release. NAG Copyright 2012.

Module d02kafe_mod

! Data for D02KAF example program

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp
Real (Kind=nag_wp), Parameter :: two = 2.0_nag_wp
Real (Kind=nag_wp), Parameter :: zero = 0.0_nag_wp
Integer, Parameter :: nin = 5, nout = 6, qq = 5

Contains
Subroutine coeffn(p,q,dqdl,x,elam,jint)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Out) :: dqdl, p, q
Real (Kind=nag_wp), Intent (In) :: elam, x
Integer, Intent (In) :: jint

! .. Intrinsic Procedures ..
Intrinsic :: cos, real

! .. Executable Statements ..
p = one

dqdl = one
q = elam - two*real(qq,kind=nag_wp)*cos(two*x)
Return

End Subroutine coeffn
Subroutine monit(nit,iflag,elam,finfo)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: elam
Integer, Intent (In) :: iflag, nit

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: finfo(15)

! .. Executable Statements ..
If (nit==14) Then

Write (nout,*)
Write (nout,*) ’Output from MONIT’

End If
Write (nout,99999) nit, iflag, elam, finfo(1:4)
Return

99999 Format (1X,2I4,F10.3,2E12.2,2F8.1)
End Subroutine monit

End Module d02kafe_mod
Program d02kafe

! D02KAF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d02kaf, d02kay, nag_wp
Use d02kafe_mod, Only: coeffn, nin, nout, one, qq, zero

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: delam, delam1, elam, elam1, tol, &

xl, xr
Integer :: i, ifail, k

! .. Local Arrays ..
Real (Kind=nag_wp) :: bcond(3,2)

! .. Executable Statements ..
Write (nout,*) ’D02KAF Example Program Results’

! Skip heading in data file
Read (nin,*)

! xl: left-hand end point, xr: right-hand end point,
! k: index of the required eigenvalue
! elam1: initial estimate of the eigenvalue
! delam1: initial search step

Read (nin,*) xl, xr
Read (nin,*) k
Read (nin,*) elam1, delam1
bcond(1,1:2) = one
bcond(2,1:2) = zero
Do i = 5, 6

tol = 10.0_nag_wp**(-i)
elam = elam1
delam = delam1

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0

! * To obtain monitoring information from the supplied
! subroutine monit replace the name d02kay by monit in
! the next statement and USE monit from d02kafe_mod *

Call d02kaf(xl,xr,coeffn,bcond,k,tol,elam,delam,d02kay,ifail)

Write (nout,*)
Write (nout,99999) ’Calculation with TOL =’, tol
Write (nout,*)
Write (nout,*) ’ Final results’
Write (nout,*)
Write (nout,99998) k, qq, elam, delam
Write (nout,99997) bcond(3,1), bcond(3,2)

Write (nout,*)
End Do

99999 Format (1X,A,E16.4)
99998 Format (1X,’ K =’,I3,’ QQ =’,I3,’ ELAM =’,F12.3,’ DELAM =’,E12.2)
99997 Format (1X,’ BCOND(3,1) =’,E12.4,’ BCOND(3,2) =’,E12.4)

End Program d02kafe

9.2 Program Data

D02KAF Example Program Data
0.0 3.14159265358979323846 : xl, xr
4 : k
15.0 4.0 : elam1, delam1

9.3 Program Results

D02KAF Example Program Results

Calculation with TOL = 0.1000E-04

Final results

K = 4 QQ = 5 ELAM = 17.097 DELAM = 0.15E-03
BCOND(3,1) = -0.9064E+00 BCOND(3,2) = 0.9064E+00

Calculation with TOL = 0.1000E-05

Final results

K = 4 QQ = 5 ELAM = 17.097 DELAM = 0.15E-04
BCOND(3,1) = -0.9075E+00 BCOND(3,2) = 0.9075E+00

___________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________

